Newsgroup: it.comp.appl.access
Titolo: [VB] Verificare se un PC (descritto su file ini esterno alla procedura) è acceso o meno

Salve,
come da oggetto, da codice, devo verificare se un determinato pc è acceso
...
esperienze in merito? (conosco il nome e l'ip e vorrei gestirlo su ini
esterno).
Saluti,
--
strano...mavero

 wrote in message
news:426f8...@x-privat.org

Pingalo.
Se esegui un Ping all'IP vedi se risponde attendendo l'ECHO...!
Io uso questo sistema per verificare sempre prima di accedere
al Server SQL se è attivo, ma è sempre tramite un'interrogazione
all'IP pertanto dovrebbe andare.
If PINGIP("xx.xx.xx.xx")="OK" then msgbox "ATTIVO"
' --------------  MODULO  ---------------------
Option Compare Database
Option Explicit
''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Copyright ©1996-2001 VBnet, Randy Birch, All Rights Reserved.
' Some pages may also contain other copyrights by the author.
''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' You are free to use this code within your own applications,
' but you are expressly forbidden from selling or otherwise
' distributing this source code without prior written consent.
' This includes both posting free demo projects made from this
' code as well as reproducing the code in text or html format.
''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
Private Const IP_STATUS_BASE As Long = 11000
Private Const IP_SUCCESS As Long = 0
Private Const IP_BUF_TOO_SMALL As Long = (11000 + 1)
Private Const IP_DEST_NET_UNREACHABLE As Long = (11000 + 2)
Private Const IP_DEST_HOST_UNREACHABLE As Long = (11000 + 3)
Private Const IP_DEST_PROT_UNREACHABLE As Long = (11000 + 4)
Private Const IP_DEST_PORT_UNREACHABLE As Long = (11000 + 5)
Private Const IP_NO_RESOURCES As Long = (11000 + 6)
Private Const IP_BAD_OPTION As Long = (11000 + 7)
Private Const IP_HW_ERROR As Long = (11000 + 8)
Private Const IP_PACKET_TOO_BIG As Long = (11000 + 9)
Private Const IP_REQ_TIMED_OUT As Long = (11000 + 10)
Private Const IP_BAD_REQ As Long = (11000 + 11)
Private Const IP_BAD_ROUTE As Long = (11000 + 12)
Private Const IP_TTL_EXPIRED_TRANSIT As Long = (11000 + 13)
Private Const IP_TTL_EXPIRED_REASSEM As Long = (11000 + 14)
Private Const IP_PARAM_PROBLEM As Long = (11000 + 15)
Private Const IP_SOURCE_QUENCH As Long = (11000 + 16)
Private Const IP_OPTION_TOO_BIG As Long = (11000 + 17)
Private Const IP_BAD_DESTINATION As Long = (11000 + 18)
Private Const IP_ADDR_DELETED As Long = (11000 + 19)
Private Const IP_SPEC_MTU_CHANGE As Long = (11000 + 20)
Private Const IP_MTU_CHANGE As Long = (11000 + 21)
Private Const IP_UNLOAD As Long = (11000 + 22)
Private Const IP_ADDR_ADDED As Long = (11000 + 23)
Private Const IP_GENERAL_FAILURE As Long = (11000 + 50)
Private Const MAX_IP_STATUS As Long = (11000 + 50)
Private Const IP_PENDING As Long = (11000 + 255)
Private Const PING_TIMEOUT As Long = 500
Private Const WS_VERSION_REQD As Long = &H101
Private Const MIN_SOCKETS_REQD As Long = 1
Private Const SOCKET_ERROR As Long = -1
Private Const INADDR_NONE As Long = &HFFFFFFFF
Private Const MAX_WSADescription As Long = 256
Private Const MAX_WSASYSStatus As Long = 128
Private Type WSADATA
   wVersion As Integer
   wHighVersion As Integer
   szDescription(0 To MAX_WSADescription) As Byte
   szSystemStatus(0 To MAX_WSASYSStatus) As Byte
   wMaxSockets As Long
   wMaxUDPDG As Long
   dwVendorInfo As Long
End Type
Private Type ICMP_OPTIONS
   Ttl             As Byte
   Tos             As Byte
   Flags           As Byte
   OptionsSize     As Byte
   OptionsData     As Long
End Type
Public Type ICMP_ECHO_REPLY
   Address         As Long
   status          As Long
   RoundTripTime   As Long
   DataSize        As Long
  'Reserved        As Integer
   DataPointer     As Long
   Options         As ICMP_OPTIONS
   Data            As String * 250
End Type
Private Type HOSTENT
   hName As Long
   hAliases As Long
   hAddrType As Integer
   hLen As Integer
   hAddrList As Long
End Type
Private Declare Function gethostbyname Lib "WSOCK32.DLL" _
  (ByVal hostname As String) As Long
 
Private Declare Sub CopyMemory Lib "kernel32" _
   Alias "RtlMoveMemory" _
  (xDest As Any, _
   xSource As Any, _
   ByVal nbytes As Long)
Private Declare Function lstrlenA Lib "kernel32" _
  (lpString As Any) As Long
Private Declare Function WSAStartup Lib "WSOCK32.DLL" _
   (ByVal wVersionRequired As Long, _
    lpWSADATA As WSADATA) As Long
 
Private Declare Function WSACleanup Lib "WSOCK32.DLL" () As Long
Private Declare Function IcmpCreateFile Lib "icmp.dll" () As Long
Private Declare Function IcmpCloseHandle Lib "icmp.dll" _
   (ByVal IcmpHandle As Long) As Long
 
Private Declare Function IcmpSendEcho Lib "icmp.dll" _
   (ByVal IcmpHandle As Long, _
    ByVal DestinationAddress As Long, _
    ByVal RequestData As String, _
    ByVal RequestSize As Long, _
    ByVal RequestOptions As Long, _
    ReplyBuffer As ICMP_ECHO_REPLY, _
    ByVal ReplySize As Long, _
    ByVal Timeout As Long) As Long
 
Private Declare Function inet_addr Lib "WSOCK32.DLL" _
   (ByVal s As String) As Long
Public Function PINGIP(ipadd as string) as string
   Dim ECHO As ICMP_ECHO_REPLY
   Dim success As Long
   Dim sIPAddress As String
 
   If SocketsInitialize() Then
     'ping the ip passing the address, text
     'to use, and the ECHO structure
      success = Ping(ipadd , ("Test"), ECHO)
     'display the results
      PINGIP = GetStatusCode(success)
      SocketsCleanup
   Else
        MsgBox "Windows Sockets for 32 bit Windows " & _
               "environments is not successfully responding."
   End If
End Function
Public Function Ping(sAddress As String, _
                     sDataToSend As String, _
                     ECHO As ICMP_ECHO_REPLY) As Long
  'If Ping succeeds :
  '.RoundTripTime = time in ms for the ping to complete,
  '.Data is the data returned (NULL terminated)
  '.Address is the Ip address that actually replied
  '.DataSize is the size of the string in .Data
  '.Status will be 0
  '
  'If Ping fails .Status will be the error code
 
   Dim hPort As Long
   Dim dwAddress As Long
 
  'convert the address into a long representation
   dwAddress = inet_addr(sAddress)
 
  'if dwAddress is valid
   If dwAddress <> INADDR_NONE Then
 
     'open a port
      hPort = IcmpCreateFile()
 
     'and if successful,
      If hPort Then
 
        'ping it.
         Call IcmpSendEcho(hPort, _
                           dwAddress, _
                           sDataToSend, _
                           Len(sDataToSend), _
                           0, _
                           ECHO, _
                           Len(ECHO), _
                           PING_TIMEOUT)
        'return the status as ping success
         Ping = ECHO.status
        'close the port handle
         Call IcmpCloseHandle(hPort)
 
      End If  'If hPort
 
   Else:
 
        'the address format was probably invalid
         Ping = INADDR_NONE
 
   End If
 
End Function
Public Function GetStatusCode(status As Long) As String
   Dim Msg As String
 
      Select Case status
      Case IP_SUCCESS:               Msg = "ip success"
      Case INADDR_NONE:              Msg = "inet_addr: bad IP format"
      Case IP_BUF_TOO_SMALL:         Msg = "ip buf too_small"
      Case IP_DEST_NET_UNREACHABLE:  Msg = "ip dest net unreachable"
      Case IP_DEST_HOST_UNREACHABLE: Msg = "ip dest host unreachable"
      Case IP_DEST_PROT_UNREACHABLE: Msg = "ip dest prot unreachable"
      Case IP_DEST_PORT_UNREACHABLE: Msg = "ip dest port unreachable"
      Case IP_NO_RESOURCES:          Msg = "ip no resources"
      Case IP_BAD_OPTION:            Msg = "ip bad option"
      Case IP_HW_ERROR:              Msg = "ip hw_error"
      Case IP_PACKET_TOO_BIG:        Msg = "ip packet too_big"
      Case IP_REQ_TIMED_OUT:         Msg = "ip req timed out"
      Case IP_BAD_REQ:               Msg = "ip bad req"
      Case IP_BAD_ROUTE:             Msg = "ip bad route"
      Case IP_TTL_EXPIRED_TRANSIT:   Msg = "ip ttl expired transit"
      Case IP_TTL_EXPIRED_REASSEM:   Msg = "ip ttl expired reassem"
      Case IP_PARAM_PROBLEM:         Msg = "ip param_problem"
      Case IP_SOURCE_QUENCH:         Msg = "ip source quench"
      Case IP_OPTION_TOO_BIG:        Msg = "ip option too_big"
      Case IP_BAD_DESTINATION:       Msg = "ip bad destination"
      Case IP_ADDR_DELETED:          Msg = "ip addr deleted"
      Case IP_SPEC_MTU_CHANGE:       Msg = "ip spec mtu change"
      Case IP_MTU_CHANGE:            Msg = "ip mtu_change"
      Case IP_UNLOAD:                Msg = "ip unload"
      Case IP_ADDR_ADDED:            Msg = "ip addr added"
      Case IP_GENERAL_FAILURE:       Msg = "ip general failure"
      Case IP_PENDING:               Msg = "ip pending"
      Case PING_TIMEOUT:             Msg = "ping timeout"
      Case Else:                     Msg = "unknown  msg returned"
   End Select
   'GetStatusCode = CStr(status) & "   [ " & msg & " ]"
    If status = 0 Then
        GetStatusCode = "OK"
    Else
        GetStatusCode = "Errore"
    End If
 
End Function
Public Function GetIPFromHostName(ByVal sHostName As String) As String
  'converts a host name to an IP address.
   Dim nbytes As Long
   Dim ptrHosent As Long  'address of hostent structure
   Dim ptrName As Long    'address of name pointer
   Dim ptrAddress As Long 'address of address pointer
   Dim ptrIPAddress As Long
   Dim sAddress As String
 
   sAddress = Space$(4)
   ptrHosent = gethostbyname(sHostName & vbNullChar)
   If ptrHosent <> 0 Then
     'assign pointer addresses and offset
 
     'ptrName is the official name of the host (PC).
     'If using the DNS or similar resolution system,
     'it is the Fully Qualified Domain Name (FQDN)
     'that caused the server to return a reply.
     'If using a local hosts file, it is the first
     'entry after the IP address.
      ptrName = ptrHosent
 
     'Null-terminated list of addresses for the host.
     'The Address is offset 12 bytes from the start of
     'the HOSENT structure. Note: Here we are retrieving
     'only the first address returned. To return more than
     'one, define sAddress as a string array and loop through
     'the 4-byte ptrIPAddress members returned. The last
     'item is a terminating null. All addresses are returned
     'in network byte order.
      ptrAddress = ptrHosent + 12
 
     'get the IP address
      CopyMemory ptrName, ByVal ptrName, 4
      CopyMemory ptrAddress, ByVal ptrAddress, 4
      CopyMemory ptrIPAddress, ByVal ptrAddress, 4
      CopyMemory ByVal sAddress, ByVal ptrIPAddress, 4
      GetIPFromHostName = IPToText(sAddress)
   End If
 
End Function
Private Function IPToText(ByVal IPAddress As String) As String
   IPToText = CStr(Asc(IPAddress)) & "." & _
              CStr(Asc(Mid$(IPAddress, 2, 1))) & "." & _
              CStr(Asc(Mid$(IPAddress, 3, 1))) & "." & _
              CStr(Asc(Mid$(IPAddress, 4, 1)))
 
End Function
Public Sub SocketsCleanup()
 
   If WSACleanup() <> 0 Then
       MsgBox "Windows Sockets error occurred in Cleanup.",
vbExclamation
   End If
 
End Sub
Public Function SocketsInitialize() As Boolean
   Dim WSAD As WSADATA
 
   SocketsInitialize = WSAStartup(WS_VERSION_REQD, WSAD) = IP_SUCCESS
 
End Function
@Alex
-- 
Posted via Mailgate.ORG Server - 
http://www.Mailgate.ORG

[cut]
Fantastico funziona ma se invece di questa stringa:
If PINGIP("xx.xx.xx.xx")="OK" then msgbox "ATTIVO"
al posto dell'ip volessi passargli un dato contenuto in un file di tipo ini
esterno come la scrivo?
Ho provato cosě ... ma mi da errore:
If PINGIP(LeggiIniFile(App.Path & "\config.ini", "RETE", "IP", "")) = "OK"
Then Me.Timer1.Enabled = True
Else: Me.Timer1.Enabled = False
COSA DOVREI SCRIVERE, O MEGLIO ... COME DOVREI SCRIVERLO?
Funzione LeggiIniFile (usata regolarmente)
Public Function LeggiIniFile(FileIni As String, Sezione As String, Campo As
String, ValDefault As String) As String
Dim Comodo As String
Dim x As Integer
    On Error GoTo Errore
    Comodo = String(2048, 32)
    x = GetPrivateProfileString(Sezione, Campo, ValDefault, Comodo,
Len(Comodo), FileIni)
    LeggiIniFile = Mid(Comodo, 1, x)
    Exit Function
Errore:
    If GestoreErrori Then Resume
End Function
Grazie !!
--
strano...mavero

 COSA DOVREI SCRIVERE, O MEGLIO ... COME DOVREI SCRIVERLO?
[CUT]
Cosa ci vorresti fare con il file.INI...?
Quel metodo apre una connessione via Socket al Server
pertanto serve l'IP, quindi gli invia una stringa qualsiasi, alla
quale il Server risponde con un messaggio compreso nell'elenco,
si chiama ECHO.
Se non c'č l'ECHO non č attivo.
A questo punto ti richiedo cosa vuoi fare con il File INI...?
Devi sempre avere l'elenco degli IP, oppure il ServerHostName
dal quale ricavare l'IP tramite la funzione GetIPFromHostName
che ti ho lasciato,  e passarlo al PING.
Non mi č chiaro.


[cut]
mi serve il file ini perchč installerň la procedura su diversi abienti e su
ciascuno di questi ci sono famiglie di indirizzi differenti e quindi ogni
installazione avrebbe il proprio file ini con il relativo IP al quale
punterei ... cosě č piů chiaro?
Scusa, ma non ho mai provato a fare ping via codice e sono un pň a digiuno!
--
strano...mavero

 wrote in message
news:426f9...@x-privat.org

Non preoccuparti, quello che non capivo era il file INI.
Quindi usi il file INI per recuperare l'indirizzo IP..?
Allora non c'č problema, peschi dall'INI l'IP e lo passi
alla funzione PINGIP.
Prima hai scritto cosě:
If PINGIP(LeggiIniFile(App.Path & "\config.ini", "RETE", "IP", "")) =
"OK"
Quě c'č un errore, nel senso che a prescindere dal
dato reso dalla funzione LeggiIniFile che come mi hai
spiegato non č altro che un IP, non hai inserito il messaggio
da inviare al Server per l'ECHO.
Prova questo:
If PINGIP(LeggiIniFile(App.Path & "\config.ini", "RETE", "IP",
""),"TEST") = "OK"
Ciao


[cut]
ho risolto così:
Me.iping.Text = LeggiIniFile(App.Path & "\config.ini", "RETE", "IP", "")
    If PINGIP(Me.iping.Text) = "OK" Then
       Me.Timer1.Enabled = True
       Me.Timer2.Enabled = False
    Else
       Me.Timer1.Enabled = False
       Me.Timer2.Enabled = True
    End If
nel caso negativo ... il timer2 ritenta ogni 5 minuti fino a quando la
macchina richiesta non è pingabile a quel punto abilito il timer1 il quale
piloterà la funzione da me richiesta.
Saluti,
--
strano...mavero

 wrote in message
news:426fa...@x-privat.org

Ottimo.
Attenzione a questo:
Me.iping.Text  
Me!iPing non mettere Text, deve essere attivo.
Poi se non ti serve visualizzare il valore IP nel controllo
usa una stringa:
strIP=LeggiInifile(....)
